# $Id: PKGBUILD 176885 2013-02-02 04:32:35Z bisson $
# Maintainer: Gaetan Bisson <bisson@archlinux.org>
# Contributor: Firmicus <francois.archlinux.org>
# Contributor: Lukas Jirkovsky <l.jirkovsky@gmail.com>
# Contributor: <AttilaH©web.de>

pkgname=ttf-linux-libertine
pkgver=5.3.0
_reldate=2012_07_02
pkgrel=2
url='http://linuxlibertine.sourceforge.net/'
pkgdesc='Serif (Libertine) and Sans Serif (Biolinum) OpenType fonts with large Unicode coverage'
license=('GPL' 'custom:OFL')
arch=('any')
depends=('fontconfig' 'xorg-fonts-encodings' 'xorg-font-utils')
provides=('ttf-font')
source=("http://downloads.sourceforge.net/project/linuxlibertine/linuxlibertine/${pkgver}/LinLibertine"{TTF,OTF}"_${pkgver}_${_reldate}.tgz")
sha1sums=('3e83241395446dcf00c7831b8b587fd55e7a7eda'
          'e0badcfff85dcfcedfd2fbb3c1c780e742cced5f')

install=install

package() {
	cd "${srcdir}"
	install -dm755 "${pkgdir}"/usr/share/fonts/{OTF,TTF}
	install -m644 *.ttf "${pkgdir}"/usr/share/fonts/TTF
	install -m644 *.otf "${pkgdir}"/usr/share/fonts/OTF

	install -Dm644 OFL-1.1.txt "${pkgdir}"/usr/share/licenses/"${pkgname}"/LICENSE
	install -Dm644 LICENCE.txt "${pkgdir}"/usr/share/licenses/"${pkgname}"/README
}
